
-------------------------------------------------------------------------------

List of Sample LINGO Solutions:

1. Multi-Period Inventory/Production Problem (Sailco)

2. Worker Scheduling Problem (Post Office)

3. Transportation Problem (Powerco)

4. Assignment Problem (Machineco)

5. Shortest Path Problem (TBD)

6. Max Flow Problem (Sunco Oil)

7. Critical Path Problem (Widgetco)

8. Critical Path/Total Float Problem Without Using LP (Widgetco2)

9. Minimum Cost Network Flow Problem (Traffic)

10. Two-Person Zero-Sum Game (Game)

-------------------------------------------------------------------------------

1. Multi-Period Inventory/Production Problem (Sailco)

-------------------------------------------------------------------------------

!  Sailco sailboat inventory problem - p. 101 (LINGO p. 163)

min z = 400x(1) + 400x(2) + 400x(3) + 400x(4) + 

        450y(1) + 450y(2) + 450y(3) + 450y(4) + 

         20i(1) +  20i(2) +  20i(3) +  20i(4) + 

s.t. x(1) <= 40,  x(2) <= 40,  x(3) <= 40,  x(4) <= 40

     i(1) =   10 + x(1) + y(1) - 40

     i(2) = i(1) + x(2) + y(2) - 60

     i(3) = i(2) + x(3) + y(3) - 75

     i(4) = i(3) + x(4) + y(4) - 25

     all x(i), y(i) and i(i) >= 0

;
MODEL:

  SETS:

    QUARTERS/1..4/:DEM,RP,OP,INV;

  ENDSETS
  MIN=@SUM(QUARTERS(I):400*RP(I)+450*OP(I)+20*INV(I));

  @FOR(QUARTERS(I):RP(I)<40);

  INV(1)=10+RP(1)+OP(1)-DEM(1);  ! for Q1;
  @FOR(QUARTERS(I)|I#GT#1:

    INV(I)=INV(I-1)+RP(I)+OP(I)-DEM(I););  ! for Q2-Q4;
  DATA:

    DEM=40,60,75,25;

  ENDDATA
END
-------------------------------------------------------------------------------

2. Worker Scheduling Problem (Post Office)

-------------------------------------------------------------------------------

! Post Office Scheduling Problem - p. 72 (LINGO p. 165)

min z = x(1) + x(2) + x(3) + x(4) + x(5) + x(6) + x(7)

        x(1)               + x(4) + x(5) + x(6) + x(7) >= 17

        x(1) + x(2)               + x(5) + x(6) + x(7) >= 13

        x(1) + x(2) + x(3)               + x(6) + x(7) >= 15
        x(1) + x(2) + x(3) + x(4)               + x(7) >= 19
        x(1) + x(2) + x(3) + x(4) + x(5)               >= 14
             + x(2) + x(3) + x(4) + x(5) + x(6)        >= 16
                    + x(3) + x(4) + x(5) + x(6) + x(7) >= 11
        all x(i) >= 0

;
MODEL:

  SETS:

    DAYS/MON,TUE,WED,THU,FRI,SAT,SUN/:RQMT,START;

  ENDSETS
  MIN=@SUM(DAYS:START);

  @FOR(DAYS(I):@SUM(DAYS(J)|(J#GT#I+2)#OR#(J#LE#I#AND#J#GT#I-5):

    START(J))>RQMT(I););

  DATA:

    RQMT=17,13,15,19,14,16,11;

  ENDDATA
END
-------------------------------------------------------------------------------

3. Transportation Problem (Powerco)

-------------------------------------------------------------------------------

! Powerco Transportation Problem - p. 364 (LINGO p368)

          City 1     City 2     City 3     City 4        Supply

Plant 1      8          6         10          9             35

Plant 2      9         12         13          7             50

Plant 3     14          9         16          5             40

Demand      45         20         30         30          

;
MODEL:

  SETS:

    PLANTS/P1,P2,P3/:CAP;

    CITIES/C1,C2,C3,C4/:DEM;

    LINKS(PLANTS,CITIES):COST,SHIP;

  ENDSETS
  MIN=@SUM(LINKS:COST*SHIP);

  @FOR(CITIES(J):

    @SUM(PLANTS(I):SHIP(I,J))>DEM(J));

  @FOR(PLANTS(I):

    @SUM(CITIES(J):SHIP(I,J))<CAP(I));

  DATA:

    CAP=35,50,40;

    DEM=45,20,30,30;

    COST= 8, 6,10, 9,

          9,12,13, 7,

         14, 9,16, 5;

  ENDDATA
END
-------------------------------------------------------------------------------

4. Assignment Problem (Machineco)

-------------------------------------------------------------------------------

! Machineco Assignment Problem - p. 394 (LINGO p398)

             Job 1      Job 2      Job 3      Job 4

Machine 1     14          5          8          7

Machine 2      2         12          6          5

Machine 3      7          8          3          9

Machine 4      2          4          6         10

                          Setup Times

;
MODEL:

  SETS:

    MACHINES/1..4/;

    JOBS/1..4/;

    LINKS(MACHINES,JOBS):COST,ASSIGN;

  ENDSETS
  MIN=@SUM(LINKS:COST*ASSIGN);

  @FOR(MACHINES(I):

    @SUM(JOBS(J):ASSIGN(I,J))=1);

  @FOR(JOBS(J):

    @SUM(MACHINES(I):ASSIGN(I,J))=1);

  DATA:

  COST = 14, 5, 8, 7,

          2,12, 6, 5,

          7, 8, 3, 9,

          2, 4, 6,10;

  ENDDATA
END
-------------------------------------------------------------------------------

5. Shortest Path Problem (TBD)

-------------------------------------------------------------------------------

-------------------------------------------------------------------------------

6. Max Flow Problem (Sunco Oil)

-------------------------------------------------------------------------------

! Sunco Oil Max Flow Problem - p. 420 (LINGO p423)

Five Nodes: so, 1, 2, 3, and si

Arc    Capacity

(so,1)     2

(so,2)     3

(1,2)      3

(1,3)      4

(3,si)     1

(2,si)     2

;
MODEL:

  SETS:

    NODES/so, 1, 2, 3, si/;

    ARCS(NODES,NODES)/so,1 so,2 1,2 1,3 2,si 3,si  si,so/

      :CAP,FLOW;

  ENDSETS
  MAX=FLOW(5,1);

  @FOR(ARCS(I,J):FLOW(I,J)<CAP(I,J));

  @FOR(NODES(I):

    @SUM(ARCS(J,I):FLOW(J,I))=@SUM(ARCS(I,J):FLOW(I,J)));

  DATA:

    CAP=2,3,3,4,2,1,1000;

  ENDDATA
END
-------------------------------------------------------------------------------

7. Critical Path Problem (Widgetco)

-------------------------------------------------------------------------------

! Widgetco Critical Path Problem - p. 434 (LINGO p441)

Activity                  Predecessors     Duration (days)

A = train workers               -              6

B = purchase raw materials      -              9

C = produce product 1          A,B             8

D = produce product 2          A,B             7

E = test product 2              D             10

F = assemble products 1 & 2    C,E            12

;
MODEL:

  SETS:

    NODES/1..6/:TIME;

    ARCS(NODES,NODES)/

      1,2 1,3 2,3 3,4 3,5 4,5 5,6/:DUR;

  ENDSETS
  MIN=TIME(6)-TIME(1);

  @FOR(ARCS(I,J):TIME(J)>TIME(I)+DUR(I,J));

  DATA:

    DUR=9,6,0,7,8,10,12;

  ENDDATA
END
-------------------------------------------------------------------------------

8. Critical Path/Total Float Problem Without Using LP (Widgetco2)

-------------------------------------------------------------------------------

! Widgetco2 Critical Path/Total Float Without Solving LP - p. 434 (LINGO p442)

Activity                  Predecessors     Duration (days)

A = train workers               -              6

B = purchase raw materials      -              9

C = produce product 1          A,B             8

D = produce product 2          A,B             7

E = test product 2              D             10

F = assemble products 1 & 2    C,E            12

;
MODEL:

  SETS:

    NODES/1..6/:ET,LT;

    ARCS(NODES,NODES)/1,2 1,3 2,3 3,4 3,5 4,5 5,6/:DUR,TFLOAT;

  ENDSETS
  DATA:

    DUR = 9,6,0,7,8,10,12;

  ENDDATA
  ET(1)=0;

  @FOR( NODES( J)| J  #GT# 1:

    ET( J) = @MAX( ARCS( I, J): ET( I) + DUR(I,J)););

  LNODE=@SIZE(NODES);

  LT( LNODE) = ET( LNODE);

  @FOR( NODES(I)| I #LT#LNODE:

    LT( I) = @MIN( ARCS( I, J): LT( J) - DUR( I,J)););

  @FOR(ARCS(I,J):TFLOAT(I,J)=LT(J)-LT(I)-DUR(I,J));

END
-------------------------------------------------------------------------------

9. Minimum Cost Network Flow Problem (Traffic)

! Traffic Minimum-Cost Network Flow Problem Example 7, p452 (LINGO p454)

Six nodes (1-6), Nine arcs.

 Arc        Time (mins)      Cars/hr

(1,2)           10             800

(1,3)           50             600

(2,4)           30             600

(2,5)           70             100

(3,4)           10             300

(3,5)           60             400

(4,5)           30             600

(4,6)           60             400

(5,6)           30             600

;
MODEL:

  SETS:

    NODES/1..6/:SUPP;

    ARCS(NODES,NODES)/1,2 1,3 2,4 2,5 3,4 3,5 4,5 4,6 5,6/

      :CAP,FLOW,COST;

  ENDSETS
  MIN=@SUM(ARCS:COST*FLOW);

  @FOR(ARCS(I,J):FLOW(I,J)<CAP(I,J));

  @FOR(NODES(I):-@SUM(ARCS(J,I):FLOW(J,I))

    +@SUM(ARCS(I,J):FLOW(I,J))=SUPP(I));

  DATA:

    COST=10,50,30,70,10,60,30,60,30;

    SUPP=900,0,0,0,0,-900;

    CAP=800,600,600,100,300,400,600,400,600;

  ENDDATA
END
-------------------------------------------------------------------------------

10. Two-Person Zero-Sum Game (Game)

! Game (Two-Finger Mora) Two-Person Zero-Sum Game, Table 27 p824 (LINGO p825)

          (1,1)    (1,2)    (2,1)    (2,2)

(1,1)       0        2       -3        0

(1,2)      -2        0        0        3

(2,1)       3        0        0       -4

(2,2)       0       -3        4        0

;
MODEL:

  SETS:

    ROWS/1..4/:X;

    COLS/1..4/;

    MATRIX(ROWS,COLS):REW;

  ENDSETS
  @FOR(COLS(J):@SUM(ROWS(I):REW(I,J)*X(I))>V;);

  @SUM(ROWS(I):X(I))=1;

  MAX=V;

  @FREE(V);

  DATA:

    REW= 0, 2,-3, 0,

        -2, 0, 0, 3,

         3, 0, 0,-4,

         0,-3, 4, 0;

  ENDDATA
END
-------------------------------------------------------------------------------

